Domänenfunktionen optimal einsetzen, Teil 1
Versionen: Access 2007, 2003, 2002/XP, 2000 und 97

Domänenfunktionen sind ein wenig bekanntes Hilfsmittel, das Ihnen häufig den Einsatz komplizierter Abfragen erspart. Mit nur einer einzigen Anweisung können Sie beispielsweise feststellen, welcher Kunde die meisten Bestellungen aufgibt, welche Artikel sich schlecht verkaufen oder bei welchem Lieferanten Sie einen Artikel am günstigsten beziehen können. Dieser dreiteilige Tipp der Woche vermittelt im ersten Teil Grundlagen zu Domänenfunktionen, zeigt Ihnen im zweiten Teil, welche Domänenfunktionen Sie in verschiedenen Bereichen Ihrer Datenbanken für unterschiedlichste Aufgaben einsetzen können und erklärt im dritten Teil den Einsatz dieser Funktionen in Abfragen, Textfeldern und VBA.

Die Bezeichnung "Domänenfunktion" hört sich zunächst etwas ungewöhnlich an, aber bei näherer Betrachtung ergibt er sich aus der Arbeitsweise dieser leistungsfähigen Funktionsgruppe: Jede Auswertung mit einer Domänenfunktion bezieht sich immer auf eine bestimmte Gruppe von Datensätzen (Domäne). Dabei kann es sich um eine einzelne Tabelle, um eine Abfrage oder eine beliebige andere Datensatzgruppe handeln, die Sie per SQL definieren. Außerdem beziehen sich Domänenfunktionen in der Regel nur auf ein einziges Feld der ausgewählten Datensätze und sie liefern immer nur ein Ergebnis. Domänenfunktionen können keine eigenständigen Aktionen wie zum Beispiel das Löschen von Datensätzen oder die Änderung von Feldinhalten ausführen. Domänenfunktionen können Sie entweder in Abfragen, Formularen und Berichten oder in VBA-Routinen nutzen, wobei die grundsätzliche Syntax immer gleich ist. Es gibt allerdings einige wichtige Unterschiede:

· Domänenfunktion in Abfragen, Formularen oder Berichten: Als Trennzeichen zwischen den Parametern wird das Semikolon verwendet, die Funktionsnamen können sowohl in Deutsch als auch in Englisch eingegeben werden, Access setzt sie aber generell in Deutsch um.

· Domänenfunktion in VBA: Als Trennzeichen zwischen den Parametern wird das Komma verwendet, die Funktionsnamen müssen in Englisch eingegeben werden.

Wenn Sie also in einer Abfrage die englische Bezeichnung der Domänenfunktion wie zum Beispiel "DSum" eingeben, konvertiert Access Ihre Eingabe automatisch in die deutsche Bezeichnung "DomSumme". Im VBA-Editor führt die deutsche Bezeichnung "DomSumme" zu einem Laufzeitfehler, da hier keine automatische Konvertierung erfolgt!

Der Aufruf von Domänenfunktionen folgt immer der gleichen Syntax:

Ergebnis = DFunktion("[Feld]","Domäne"[,"Kriterien"])

Der erste Parameter "Feld" definiert das Feld, für das eine Domänenfunktion ausgeführt werden soll. Bei einigen Domänenfunktionen, bei denen der Feldinhalt keine Rolle spielt, können Sie an dieser Stelle auch ein "*" (Sternchen) angeben. Der zweite Parameter "Domäne" legt die Domäne fest und damit die Datensätze, die Sie auswerten möchten. Bei einer Domäne kann es sich um eine Tabelle, eine Abfrage oder einen SQL-Ausdruck handeln. Der dritte und letzte, optionale Parameter "Kriterien" bestimmt die Auswahlkriterien für die Datensätze, die aus der "Domäne" ausgewertet werden sollen.

Wenn Sie kein Kriterium angeben, bezieht sich die Domänenfunktion immer auf alle Datensätze der Domäne. Handelt es sich dabei beispielsweise um eine Tabelle "Bestellungen", greift die Domänenfunktion auf alle Datensätze dieser Tabelle zu. Möchten Sie die Domänenfunktion hingegen nur auf bestimmte Datensätze anwenden, legen Sie das über ein entsprechendes Kriterium fest. Dabei gilt grundsätzlich, dass Sie dieselbe Syntax verwenden müssen, wie bei der "Where-Klausel" einer SQL-Abfrage:

[Feld] <Bedingung> <Vergleich>

Zum Beispiel:

[PLZ] = 80911

Dieses Kriterium würde dafür sorgen, dass nur die Datensätze einer Tabelle selektiert werden, bei denen das Feld "PLZ" den Inhalt "80911" aufweist.

Um nun zum Beispiel mehrere Kunden aus einem bestimmten Nummernkreis zu selektieren, können Sie einerseits für die Bedingung andere Operatoren einsetzen und andererseits Kriterien kombinieren:

[KundenNr] >= 20000 AND [KundenNr] <= 29999

Diese Konstruktion selektiert alle Kunden, deren Kundennummer größer oder gleich 20000 UND kleiner oder gleich 29999 ist, die also im 20.000er-Bereich liegen. Was aber, wenn Sie nun alle 20.000er-Kunden und zusätzlich Sonderkunden ab 80.000 selektieren möchten? Hier käme dann die folgende Konstruktion zum Einsatz:

([KundenNr] >= 20000 AND [KundenNr] <= 29999)
OR ([KundenNr] >= 80000)

In diesem Fall werden die Bedingungen zunächst mit runden Klammern gruppiert und dann mit "OR" (oder) verknüpft. Ein anderes Beispiel:

([KundenNr] >= 20000 AND [KundenNr] <= 29999)
AND NOT ([KundenNr] >= 24000 AND [KundenNr] <= 23999)

Über "NOT" werden hier aus den bereits selektierten 20.000er-Kunden die Kunden ausgeschlossen, die eine Nummer im 24.000er-Bereich haben.

Beim Zusammensetzen der Kriterien müssen Sie beachten, dass beim Vergleich mit Textfeldern der zu überprüfende Begriff immer in einfachen Hochkommas eingeschlossen wird! Fehlen sie, versucht Access immer einen numerischen Vergleich durchzuführen, was bei Zeichenketten natürlich nicht funktioniert. Um beispielsweise alle Kunden zu selektieren, die in München wohnen, würde das Kriterium wie folgt aussehen:

[Ort] = 'München'

Für Textfelder gibt es außerdem noch einen besonderen Operator: "Like". Mit ihm können Sie Datensätze selektieren, bei denen der Inhalt des zu prüfenden Feldes dem Suchbegriff "ähnlich" ist:

[Kundenname] like 'Schmi*'

Dieses Kriterium selektiert alle Kunden, deren Namen mit "Schmi" beginnen, das folgende Sternchen "*" bedeutet, dass die nachfolgenden Zeichen nicht geprüft werden müssen. Das Ergebnis wären somit Kunden mit den Namen "Schmidt", "Schmitt" oder "Schmiedbauer". Analog dazu kann das Sternchen auch am Anfang oder innerhalb des Begriffs verwendet werden. Alle Kunden, deren Namen mit "bauer" enden, erhalten Sie mit:

[Kundenname] like '*bauer'

Wollen Sie aber alle Kunden auswählen, die "bauer" am Anfang, am Ende oder mitten im Namen haben, sieht das Kriterium folgendermaßen aus:

[Kundenname] like '*bauer*'

Zwei verschiedene Ersetzungszeichen stehen Ihnen zur Verfügung: Während das Sternchen dafür sorgt, dass alle führenden oder folgenden Zeichen nicht weiter geprüft werden, können über das Fragezeichen einzelne, nicht zu prüfende Zeichen festgelegt werden:

[Kundenname] like 'm??er'

würde zum Beispiel alle "Meier", "Maier", "Meyer" oder "Mayer" selektieren - im Klartext: Vorne muss der Name "M" haben, hinten "er", welche beiden Zeichen dazwischen stehen, ist egal.

Soviel zu den Grundlagen zu Domänenfunktionen. Im zweiten Teil des Tipps in der nächsten Ausgabe stellen wir Ihnen dann alle verfügbaren Domänenfunktionen mit kleinen Beispielen vor.

Im ersten Teil dieses Tipps der Woche hatten wir die Grundlagen von Domänenfunktionen und deren Syntax erläutert. In diesem Teil folgt nun ein Überblick der zur Verfügung stehenden Domänenfunktionen mit kleinen Beispielen. Neben dem deutschen Funktionsnamen finden Sie in Klammern den englischen Namen, den Sie in VBA-Prozeduren verwenden müssen. Die Syntax folgt den Konventionen für eine Verwendung als Formel in Abfragen oder ungebundenen Textfeldern, als Trennzeichen zwischen den Parametern ist dabei jeweils das Semikolon angegeben, das Sie in VBA durch das Komma ersetzen:

DomAnzahl (DCount) - Anzahl von Datensätzen ermitteln
Diese Domänenfunktion ermittelt die Anzahl von Datensätzen, die gemäß den übergebenen Kriterien aus Tabellen oder über Abfragen selektiert werden. Das Funktionsergebnis ist vom Datentyp "Zahl". Mit der Funktion können Sie beispielsweise die Anzahl Ihrer französischen Kunden ermitteln:

=DomAnzahl("*"; "Kunden"; "Land = 'Frankreich'")

DomMin, DomMax (DMin, DMax) - Höchsten/niedrigsten Wert ermitteln
Mit diesen Funktionen ermitteln Sie den höchsten oder niedrigsten Wert eines Feldes in den selektierten Datensätzen. Um zum Beispiel die nächste freie Artikelnummer zu ermitteln, käme die folgende Anweisung zum Einsatz:

=DomMax("[Artikel-Nr]"; "Artikel")+1

Die Funktion ermittelt zunächst die höchste, bislang vergebene Artikelnummer und addiert dann "1" hinzu. Wie Sie sehen, muss der dritte Parameter, das Kriterium für die Selektion, nicht immer angegeben werden. Fehlt der Parameter, verwendet Access immer alle Datensätze der betreffenden Tabelle oder Abfrage.

DomMittelwert (DAvg) - Mittelwerte berechnen
Mit dieser Funktion können Sie den Mittelwert aller Werte eines Feldes in den selektierten Datensätzen ermitteln. Die Anweisung

=DomMittelwert("[Einzelpreis]"; "Artikel")

verrät Ihnen beispielsweise, wie teuer Ihre Artikel im Durchschnitt sind. Bitte beachten Sie, dass DomMittelwert() nur mit Feldern vom Datentyp "Zahl" oder "Währung" eingesetzt werden kann!

DomStdAbw, DomStdAbwG (DStDev, DStdDevP) - Standardabweichungen ermitteln
Diese Funktionen werden eingesetzt, um Standardabweichungen einer Gruppe von Werten in den selektierten Datensätzen zu ermitteln. DomStdAbw() wertet eine Population aus, DomStdAbwG() hingegen eine Populationsstichprobe. Die folgende Anweisung liefert die Standardabweichung der Frachtkosten für Lieferungen nach England:

=DomStdAbw("[Frachtkosten]"; "Bestellungen"; 
"[Bestimmungsland] = 'Großbritannien'")

DomSumme (DSum) - Summen berechnen
Mit dieser Funktion können Sie die Inhalte von Feldern der selektierten Datensätze aufsummieren und so beispielsweise Gesamtsummen aus Bestellungen für einzelne Artikel anzeigen lassen:

=DomSumme("[Anzahl]"; "Bestelldetails"; "[Artikel-Nr]= 2")

DomVarianz, DomVarianzG (DVar, DVarP) - Varianzen ermitteln
Diese Funktionen werden eingesetzt, um Varianzen in einer Gruppe von Werten in den selektierten Datensätzen zu ermitteln. DomVarianz() wertet eine Population aus, DomVarianzG() hingegen eine Populationsstichprobe. Die folgende Anweisung liefert die Varianz in den Frachtkosten für Lieferungen nach Frankreich:

DomVarianz("[Frachtkosten]"; "Bestellungen"; 
"[Bestimmungsland] = 'Frankreich'")

DomWert (DLookup) - Suche nach bestimmten Feldinhalten
Mit DomWert() können Sie gezielt den Inhalt eines Feldes aus einem bestimmten Datensatz lesen und so zum Beispiel das Ergebnis als Vorgabe in einem Formular verwenden. Das folgende Beispiel übernimmt eine neu eingegebene Postleitzahl, sucht nach einem passenden Ort und gibt ihn im Formular vor:

Sub PLZ_AfterUpdate ()
    Dim strF, LU As Variant

    On Error Resume Next
    Err = 0
    strF = Me.[PLZ]
    If Err <> 0 Then Exit Sub 'Keine Eingabe
    Err = 0
    LU = DLookup("[Ort]", _
                 "Kunden", _
                 "[PLZ] = '" & strF & "'")
    If Err <> 0 Then Exit Sub 'Nicht gefunden
    Me.[Ort] = LU

End Sub

Die Routine ist der Ereignis-Prozedur "Nach Aktualisierung" des Feldes "PLZ" zugeordnet und wird aktiviert, wenn der Anwender eine neue Postleitzahl eingibt oder eine vorhandene ändert. Das Funktionsergebnis von DomWert() entspricht immer dem Datentyp des Feldes, das Sie ausgelesen haben.

DomErsterWert, DomLetzterWert (DFirst, DLast) - Ersten/letzten Wert ermitteln
Mit diesen Funktionen können Sie gezielt auf Feldinhalte der selektierten Datensätze zugreifen, wobei Access den Feldinhalt des jeweils ersten (DomErsterWert) oder letzten (DomLetzterWert) passenden Datensatzes liefert.

Bitte beachten Sie: "DomErsterWert" und "DomLetzterWert" beziehen sich nicht auf den ersten oder letzten erfassten Datensatz! Der Einsatz von DomLetzterWert() und DomErsterWert() ist nur sinnvoll mit indizierten Tabellen oder sortierten Abfragen, andernfalls erhalten mehr oder weniger zufällige Ergebnisse!

Die folgende Konstruktion basiert auf einer Abfrage, die die Bestellungen aufsteigend nach Datum sortiert:

=DomErsterWErt("[Bestelldatum]"; "Bestellungen"; "[Kunden-
Code]= 'ALFKI'")

=DLast("[Bestelldatum]"; "Bestellungen"; "[Kunden-Code]= 
'ALFKI'")

DomErsterWert() liefert hier als Ergebnis das Datum der ersten, jemals getätigten Bestellung des Kunden mit dem Kunden-Code "ALFKI", DomLetzterWert() analog dazu das Datum der zuletzt getätigten Bestellung des Kunden.

Soviel zu den generell verfügbaren Domänenfunktionen. Dem wichtigen Einsatzgebiet von Domänenfunktionen in Abfragen oder Textfeldern widmet sich der dritte Teil dieses Tipps der Woche in der nächsten Ausgabe von Access Weekly.

In den ersten Teilen dieses Tipps der Woche hatten wir die Grundlagen sowie die generell verfügbaren Domänenfunktionen erläutert. In dieser Woche geht es nun um die wichtigen Einsatzgebiete von Domänenfunktionen in Abfragen, Textfeldern und VBA.

Nehmen wir an, dass Sie eine Kundenliste erstellen möchten, in der neben Name und Ort des Kunden die bisherigen Umsätze aufgelistet werden sollen. Die Basis bildet die Tabelle "Kunden", aus der zunächst die Felder "Firma" und "Ort" stammen. Ein drittes Feld "Umsatz" wird dynamisch generiert und beinhaltet die Formel:

Umsatz: Format(DomSumme("[Endpreis]"; 
"Umsatz/Bestellungen"; 
"[Kunden-Code]= '" & [Kunden-Code] & "'");"#.###,00")

Die Domänenfunktion "DomSumme()" greift auf die Abfrage "Umsatz/Bestellungen" zu, die alle Umsätze für alle Kundenbestellungen selektiert. Anhand des Kriteriums

[Kunden-Code]= '" & [Kunden-Code] & "'"

wird pro Kunde der bisher getätigte Umsatz aufsummiert und dem Abfragefeld "Umsatz" zugewiesen. So lässt sich eine weitere Abfrage einsparen, die andernfalls für die Summierung pro Kunden notwendig gewesen wäre.

Außer in Abfragen kommen Domänenfunktionen in ungebundenen Textfeldern zum Einsatz. Dazu ein einfaches Beispiel: In einem Formular für Bestellungen möchten Sie in einem zusätzlichen Feld anzeigen, wie viele Bestellungen der aktuelle Kunde bislang aufgegeben hat. Dazu fügen Sie zunächst ein ungebundenes Textfeld in das Formular ein und weisen ihm in der Eigenschaft "Steuerelementeinhalt" den folgenden Ausdruck zu:

=DomAnzahl("*"; "Bestellungen"; "
[Kunden-Code]= '" & [Kunden-Code] & "'")

Das Gleichheitszeichen sorgt zunächst dafür, dass das Ergebnis der Domänenfunktion im Textfeld angezeigt wird. "DomAnzahl" ist der Name der Funktion, die ausgeführt werden soll. Diese Funktion zählt die Datensätze, die gemäß der beiden letzten Parameter selektiert wurden. Wie erwähnt, erwartet eine Domänenfunktion als ersten Parameter immer den Feldnamen. Wenn, wie hier beim Zählen von Datensätzen, kein Feldinhalt ausschlaggebend ist, setzen Sie am besten ein Sternchen "*" als Feldnamen ein.

Der zweite Parameter "Bestellungen" legt den Namen der Tabelle fest, in der die Datensätze zu zählen sind. Weil aber nur die Bestellungen des im Formular angezeigten Kunden ausschlaggebend sind, müssen Sie über den dritten Parameter ein entsprechendes Kriterium übergeben. In diesem Fall ist das der Ausdruck:

[Kunden-Code] = "'" & [Kunden-Code] & "'"

Dabei bezieht sich das erste "[Kunden-Code]" auf das Feld in der Tabelle und das zweite "[Kunden-Code]" auf das Feld im Formular. Der aktuelle Feldinhalt im Formular wird hier mit dem Operator "&" in den Ausdruck "eingebaut". Bitte beachten Sie, dass einfache Hochkommas eingesetzt werden müssen, da das Feld "Kunden-Code" in der Tabelle vom Datentyp "Text" ist!

Zum Abschluss noch ein paar Worte zum Aufruf von Domänenfunktionen per Schaltfläche: Das Ergebnis einer Domänenfunktion lässt sich ohne VBA über eine Schaltfläche abrufen. Zum Beispiel: Per Mausklick auf eine Schaltfläche möchten Sie den Umsatz des aktuellen Kunden ermitteln. Dazu legen Sie im Formular eine Schaltfläche UMSATZ? an und weisen der Eigenschaft "Beim Klicken" die folgende Formel zu:

=Meldung(DomSumme("[Endpreis]"; "Umsatz/Bestellungen"; 
"[Kunden-Code]= '" & [Kunden-Code] & "'"))

Das Ergebnis ist hier nicht in einem Textfeld zu sehen, sondern wird über die Anweisung "Meldung()" in einem kleinen Fenster angezeigt. Das hat den Vorteil, dass die Daten nicht permanent im Formular aktualisiert werden müssen, sondern nur bei Bedarf per Klick auf die Schaltfläche abgerufen werden.

Beim Einsatz der eben gezeigten Lösung ist Ihnen sicherlich aufgefallen, dass das angezeigte Ergebnis nicht perfekt formatiert ist. Beträge werden ohne Tausendertrennung und teilweise nur mit einer Nachkommastelle angezeigt. Um eine entsprechende Formatierung vorzunehmen, binden Sie die Domänenfunktion in eine VBA-Routine ein, die über die Ereignis-Prozedur "Beim Klicken" der Schaltfläche aktiviert wird:

Private Sub btnUmsatzFormatiert_Click()
    Dim dblU As Double, strK As String, strMsg As String

    strK = Me.[Kunden-Code].Column(0)
    dblU = DSum("[Endpreis]", _
                "[Umsatz/Bestellungen]", _
                "[Kunden-Code]= '" & _
                Me.[Kunden-Code] & "'")
    strMsg = "Der bisherige Umsatz des Kunden '" & _
             strK & "'" + vbCrLf
    strMsg = strMsg + "beträgt " & _
             Format$(dblU, "#,###.00") & " €"
    MsgBox strMsg

End Sub

In dieser Ereignisprozedur wird zunächst der Name des aktuellen Kunden aus dem Formular ausgelesen und in der Variablen "strK" festgehalten. Die folgende Zeile weist dann das Ergebnis der Domänenfunktion "DSum()" der Variablen "dblU" zu. Abschließend erfolgt die Anzeige per "MsgBox()", wobei eine Formatierung der Summe über die Funktion "Format()" eingebaut ist.

